

DYNAMICS OF THE HAY AGGREGATOR FELLA TS1602 HYDRO

Šmíd V., Bartoň S.

Department of Engineering and Automobile Transport, Faculty of Agronomy, Mendel University of Agriculture and Forestry in Brno, Zemedelska 1, 613 00 Brno, Czech Republic

E-mail: vlasmi@centrum.cz

ABSTRACT

The work presents mathematical modelling of working cycle of hay aggregator. Analytical model is based on the programme Maple 11. It leads to the evaluation of the influence of dimensions of constructional parts and working parameters on dynamics and power consumption during one working cycle. It also enables determination of the power consumption of the whole machine. This approach enables creation of precise graphs showing kinematics of the working parts as well as computation and drawing of graphical dependencies showing dynamic stress or energy consumption. The model is a powerful tool for predicting of fuel consumption dependence and operational wear of working parts.

Key words: Maple 11, mathematical modelling, stroke area, working curve, power, force.

Acknowledgments: Authors would like to express their grateful thanks to the company TOKO-AGRO a.s. Rudice u Luhačovic for kind lending of the constructional parts of the hay aggregator

ÚVOD

Trvalé travní porosty a pícniny na orné půdě zaujímají stále významné místo v našem i celosvětovém zemědělství i když celková plocha osetá pícninami se stále zmenšuje z 1 099 tis. ha pícnin na orné půdě v roce 1990 na 513 tis. ha v roce 2003. Tento trend souvisí s útlumem živočišné výroby v našem zemědělství.

Rozloha luk a pastvin se naopak postupně zvyšuje. Tak např. v roce 1990 bylo v České republice asi 833 tis.. ha luk a pastvin a v roce 2003 bylo již 971 tis. ha trvalých travních porostů.

Výhledově je možné předpokládat, že v nejbližších letech budou oba naznačené vývojové trendy rozlohy těchto plodin pokračovat (pokles rozlohy pícnin pěstovaných na orné půdě a nárůst rozlohy luk a pastvin). Celková rozloha pícnin se proto bude měnit ve prospěch trvalých travních porostů. Nicméně i ty je nutné v rámci krajinnotvorné funkce zemědělství udržovat.

Pícniny jsou velmi důležitou plodinou z hlediska zajištění krmivové základny hospodářských zvířat. Jsou základním zdrojem objemných krmiv. Tvoří je travní porosty z luk a pastvin, dále víceleté pícniny – jeteloviny (vojtěška, jetel), pícní trávy a jednoleté pícniny (zvláště kukuřice a směsky) pěstované na orné půdě.

Skližeň pícnin probíhá téměř pro celé vegetační období se špičkami v době prvních sečí – senoseče (květen až červen) a sklizně silážních plodin (podle technologické zralosti). Hlavním problémem při sklizni je zmenšit riziko počasí a tím snížit sklizňové a konzervační ztráty. Nejjistější metodou je maximálně zkrátit sklizňový postup. Při špatném počasí a nevhodném způsobu sklizně, odrolem, nesebráním, nevhodnou konzervací mohou činit ztráty sušiny na hmotě 15 až 35 %, ztráty živin až 50 % a vitamínů až 100 %. Vhodným sklizňovým pracovním postupem a konzervací lze snížit riziko počasí a zabránit znehodnocení píce v průběhu uskladnění.

Porosty je vhodné sklízet v optimální technologické zralosti, v době, kdy obsah živin a vitamínů je maximální. Podle druhu píce a účelu jejího použití je doba sklizně například u vojtěšky na začátku květu, u jetele lučního červeného před začátkem kvetení, u lučních travních porostů v období od počátku metání do počátku kvetení převládajících trav. Ze vztahu mezi dobou sklizně v optimální vegetační fázi a koeficientem počasí vyplývá, že je nutno sklizeň jedné seče provést za 21 kalendářních dnů, z čehož je asi 10 pracovních dnů vhodných pro sklizeň. Pícniny je nutno sklízet nejen v oblastech rovinatých a se svahy do 12 stupňů, ale i v podhorských a horských se svahovitostí do 22 až 25 stupňů.

Vlastnosti porostů pícnin jsou z hlediska sklizňových operací velmi rozmanité, kolísají v dost širokých mezích a mnohé se mění se změnou vlhkosti pícní hmoty. Výnos zelené hmoty při jedné seči u tenkostěbelných pícnin (jetel, vojtěška, trávy, směsky, luční porost) je 15 až 50 t . ha⁻¹ při obsahu sušiny 15 až 40 %. Hustota porostu na metr čtvereční se pohybuje od 1 000 až 1 300 lodyh vojtěšky, u jetele lučního 750 až 4 000 lodyh a do 3 000 až 15 000 stébel u travních porostů. Výška rostlin u tenkostěbelných pícnin je 150 až 1 500 mm. Tloušťka stébel ve výši řezu je u tenkostěbelných pícnin 0,7 až 12 mm (z toho vojtěška 2,5 až 5,4 mm, jetel 2,8 až 6,2 mm, tráva 0,7 až 3,1 mm).

Výška sečení má vliv na velikost současné i následující sklizně a pohybuje se u tenkostěbelných pícnin v rozmezí 30 až 80 mm (trvalé luční porosty 30 až 40 mm, dočasné travní

porosty 40 až 50 mm, vojtěška, jetel, jetelotrávy 50 až 60 mm, směsky 60 až 80 mm). Příliš nízké sečení (pod 30 mm) působí negativně na obrůstání jetelovin a trav a na výnosy následující seče. Při vyšším sečení, než je uvedeno, se snižuje výnos (10 mm výšky porostu v přízemní vrstvě představuje 300 až 400 kg sušiny na hektar), zvyšuje se hromadění stařiny, zhoršuje se další etapa sklizně, například sbírání pícní hmoty a ztěžuje se další seč.

Seční tenkostěbelných pícnin lze provádět na řádky – pokosy nebo na široko. Šířka řádků – pokosů je jeden až dva metry, výška řádků u zelené i zavadlé píče je 100 až 250 mm. Při sečení na široko nebo po rozhození řádků na široko je hmotnost rozprostřené píče u zelené hmoty 1,5 až 5 kg . m², u zavadlého materiálu 0,6 až 1,8 kg . m², u suchého materiálu (sena) 0,4 až 1,1 kg . m². Šířka řádků shrnutého suchého nebo zavadlého materiálu je 0,6 až 2 m, výška řádku 0,15 až 0,8 m a hmotnost jednoho metru řádku zavadlé píče (nad 30 % sušiny) do 8 kg podle záběru shrnovače. Výnosy sena tenkostěbelných pícnin z jedné seče se pohybují podle druhu pícniny a podle toho, o kolikátou seč jde, v rozmezí tři až 10 t . ha⁻¹.

Základními články technologické linky pro sklizeň píče jsou žací stroj s případnou úpravou pokosu – obračeč dle potřeby – shrnovač – sběrací vůz, lis, nebo řezačka.

I když se to zdá jako samozřejmost, tak v rámci sklizňových prací je shrnutí zavadlé hmoty na řádek důležitým krokem. Pro maximální využití optimálního obsahu sušiny musí práce probíhat s velkou výkonností umožňující souběžný provoz shrnovače a řezačky (nebo jiného sklizňového stroje). Rozhodující faktory pro práci shrnovače jsou šířka záběru a s tím spojené množství hmoty spolu s charakterem ukládaného řádku. Shrnování píče do objemných řádků umožňuje i při nižší pojezdové rychlosti optimálně využít instalovaného výkonu moderních lisů, senážních návěsů a řezaček. Menší počet řádků s větším množstvím píče snižuje počet přejezdů těchto strojů na pozemku a podíl ztrátových časů pro otáčení.

Je zřejmé, že šířka odkládaného řádku by měla odpovídat záběru sběrače následného stroje. Méně známé je ale to, že šířka řádku by měla být v tomto intervalu co největší, aby se dosáhlo rovnoměrného zatížení plnicího ústrojí.

U shrnovačů by mělo být účinné kopírování terénu podvozky rotorů a precizní nastavení výšky by mělo sloužit především k tomu, aby nedocházelo k poškozování travního drnu a zvýšenému znečišťování píče. Kvalitního kopírování terénu je dosaženo kardanovým zavěšením rotorů. Každý rotor se přizpůsobí terénu nezávisle jak v podélném, tak i příčném směru.

Dvourotorové shrnovače se středovým ukládáním hmoty jsou známé tím, že vytvářejí velmi rovnoměrné řádky s nízkým znečištěním díky krátké vzdálenosti příčného posunu hmoty. Se dvěma rotory se lze vyšplhat do záběru jen něco přes 8 metrů, takže podmínkou využití výkonu řezačky je dostatečný výnos hmoty. Naproti tomu dvourotorové shrnovače s bočním ukládáním řádků dovolují dvě nebo více jízd složit do jednoho řádku. V tomto případě lze v jednotlivých sečích formovat řádky adekvátní sklizňové technice bez přímé závislosti na výnosu hmoty.

V současnosti se v podnicích nejvíce vyskytují i uplatňují širokozáběrové dvojrotorové shrnovače s bočním nahrnováním, které umožňují při malém výnosu hmoty ve dvou protiběžných jízdách nahrnout 2 řádky na sebe a tím dostatečně vyčíst odvozový prostředek.

MATERIÁL A METODIKA

Proto se ve své práci zabývám matematickým modelem dynamiky jednoho ramene shrnovače píce při jeho aktivní půlotáčce. Jde o stroj FELLA TS 1602 Hydro, typický zástupce této kategorie. Jedná se o výrobek firmy FELLA – Werke GmbH & Co. KG, Feucht, Deutschland. Je to tažený dvourotorový stranový shrnovač s nastavitelným záběrem 7,0 – 8,4 m umožňující shrnování do jednoho nebo dvou řádků. Rotory mají průměr 4,2 m a jsou vybaveny 12 shrnovacími rameny, shrnovací výška je nezávisle nastavitelná mechanicky klikou. Každý rotor má vlastní čtyřkolový podvozek. Natáčení shrnovacích ramen do a z pracovní polohy je zajištěno pohybem vodících rolen v křivkové dráze ze zušlechtné a ořezovací litiny. Stroj je v zadní části vybaven tandemovou nápravou řízenou od výkyvné ojové hlavy prostřednictvím otočného hřídele uloženého v hlavním nosníku. Potřebný výkon tažného prostředku 60 kW, otáčky vývodového hřídele 540 min^{-1} celková hmotnost 1950 kg.

Tato práce navazuje na detailní matematický model kinematiky shrnovače vypracovaný Šmídem a Bartoněm (2008). Proto bude využívat stejné vstupní hodnoty konstrukčních parametrů a bude opět využito programu Maple, tentokrát již verze 13. Výše uvedený model proto zde již nebude dále diskutován.

Vzhledem k rozsáhlým tvarům analytických výrazů odvozených v průběhu výpočtu použijeme následující konstrukční substitute, které korespondují s reálnými konstrukčními parametry.

r	=	délka vačky – 0,1 m
R	=	základní poloměr vodící křivkové dráhy – 0,5 m
T	=	perioda rotace – 1 s^{-1}
L	=	délka shrnovacího ramene – 1,5 m
l	=	délka shrnovacího prstu – 0,4 m
V	=	pojezdová rychlost soupravy – 10 km h^{-1}
Mt	=	hmotnost shrnovacího ramene – 5 kg
Mp	=	hmotnost shrnovacího prstu – 1 kg
dp	=	vzájemná rozteč shrnovacích prstů – 0,1 m
n	=	počet shrnovacích prstů – 4
σ	=	plošná hustota píce – 15 kg m^{-2}
g	=	tíhové zrychlení – $9,81 \text{ m s}^{-2}$

> **restart; with(plots):**

> **KSu:=[r=1/10,R=1/2,T=1,L=3/2,l=4/10,V=-120/36,Mt=5,Mp=1,dp=1/10,
n=4,sigma=15,g=9.81];**

$$KSu := \left[r = \frac{1}{10}, R = \frac{1}{2}, T = 1, L = \frac{3}{2}, l = \frac{2}{5}, V = -\frac{10}{3}, Mt = 5, Mp = 1, dp = \frac{1}{10}, n = 4, \sigma = 15, g = 9.81 \right]$$

Stanovení velikosti plochy záběru a jejího časového průběhu

Znalost průběhu plochy záběru je nejdůležitějším parametrem při následných výpočtech. Pracovní plocha je vytýčena prvním a čtvrtým prstem, její počátek sklopením prstů do pracovní polohy (souřadnice z dosáhne svého minima) a její konec vyzdvižením prstů z pracovní polohy (souřadnice z opustí své minimum). Pro veškeré další výpočty se bude předpokládat že souřadnice z je ve svém minimu a proto ji můžeme zanedbat a dále budeme pracovat pouze s x a y souřadnicí polohy prstu. Použijeme vztahy odvozené Šmídem a Bartoňem (2008)

```
> Prst1 := [-1*cos(2*Pi/T*t)*sin(arccos(z(t)/r)+1/4*Pi)
+L*sin(2*Pi/T*t)+(R+r)*cos(2*Pi/T*t),
V*t-1*sin(2*Pi/T*t)*sin(arccos(z(t)/r)+1/4*Pi)-L*cos(2*Pi/T*t)
+(R+r)*sin(2*Pi/T*t),1*cos(arccos(z(t)/r)+1/4*Pi)];
```

$$\begin{aligned} Prst1 := & \left[-l \cos\left(\frac{2\pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) + L \sin\left(\frac{2\pi t}{T}\right) + (R+r) \cos\left(\frac{2\pi t}{T}\right), \right. \\ & Vt - l \sin\left(\frac{2\pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) - L \cos\left(\frac{2\pi t}{T}\right) + (R+r) \sin\left(\frac{2\pi t}{T}\right), \\ & \left. l \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \right] \end{aligned}$$

```
> Prst4 := [-1*cos(2*Pi/T*t)*sin(arccos(z(t)/r)+1/4*Pi)
+(L-3*dp)*sin(2*Pi/T*t)+(R+r)*cos(2*Pi/T*t),
V*t-1*sin(2*Pi/T*t)*sin(arccos(z(t)/r)+1/4*Pi)
-(L-3*dp)*cos(2*Pi/T*t)+(R+r)*sin(2*Pi/T*t),
1*cos(arccos(z(t)/r)+1/4*Pi)];
```

$$\begin{aligned} Prst4 := & \left[-l \cos\left(\frac{2\pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) + (L-3 dp) \sin\left(\frac{2\pi t}{T}\right) + (R+r) \cos\left(\frac{2\pi t}{T}\right), \right. \\ & Vt - l \sin\left(\frac{2\pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) - (L-3 dp) \cos\left(\frac{2\pi t}{T}\right) + (R+r) \sin\left(\frac{2\pi t}{T}\right), \\ & \left. l \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \right] \end{aligned}$$

```
> P1:=subs(KSu,eval(subs(z(t)=-r/sqrt(2),Prst1)[1..2]));
```

```
> P4:=subs(KSu,eval(subs(z(t)=-r/sqrt(2),Prst4)[1..2]));
```

$$P1 := \left[\frac{3}{2} \sin(2\pi t) + \frac{3}{5} \cos(2\pi t), -\frac{10t}{3} - \frac{3}{2} \cos(2\pi t) + \frac{3}{5} \sin(2\pi t) \right]$$

$$P4 := \left[\frac{6}{5} \sin(2\pi t) + \frac{3}{5} \cos(2\pi t), -\frac{10t}{3} - \frac{6}{5} \cos(2\pi t) + \frac{3}{5} \sin(2\pi t) \right]$$

Dále je nutné stanovit časové okamžiky počátku a konce záběru a dobu jeho trvání. Z důvodů přehlednosti následujících grafů se posuneme o jednu pracovní periodu – trvání jedné otáčky zpět.

```
> T2:=solve(diff(P1[1],t),t); T1:=subs(KSu,simplify(T2-T/2));
```

$$T2 := \frac{1}{2} \frac{\arctan\left(\frac{5}{2}\right)}{\pi} \quad T1 := \frac{1}{2} \frac{-\arctan\left(\frac{5}{2}\right) + \pi}{\pi}$$

```
> t1:=evalf(subs(KSu,T1-T)); t2:=evalf(subs(KSu,T2-T)); d_t:=t2-t1;
```

$$t1 := -1.310559471 \quad t2 := -0.8105594708 \quad d_t := 0.5000000002$$

Nyní je možné stanovit velikost a časový průběh plochy záběru. Pro výpočet plochy, jejíž obvod tvoří uzavřená křivka C použijeme Greenovu větu.

$$S = \oint_C x dy - y dx,$$

kteřou zjednodušíme na výpočty dílčích ploch. Postup výpočtu nejlépe znázorní Obrázek 1., který vyrobíme za pomoci Maple.


```
> Tt:=seq(t1+(t2-t1)*i/100,i=0..100);
```

```

> G0:=plot([P1[],t=t1..t2],[P4[],t=t1..t2],color=[red,blue],
  thickness=3):
> G1:=polygonplot([seq(evalf([subs(t=Tt[j],[P1,P4)]],
  subs(t=Tt[j+1],[P4,P1]])),j=1..100)],
  color="DarkGreen",style=patchnograd):
> G2:=polygonplot([evalf(subs(t=t1,[P1[1],4.5],
  [P4[1],4.5],P4,P1)),evalf(subs(t=t2,[P1[1],4.5],[P4[1],4.5],
  P4,P1))]),color="LightGrey"):
> G3:=polygonplot(evalf([subs(t=t1,[P4[1],4.5]),
  seq(subs(t=Tt[j],P4),j=1..101),subs(t=t2,[P4[1],4.5])]),
  color="LightBlue"):
> G3a:=polygonplot([[0,4.5],[1,4.5],[1,3.5],[0,3.5]],
  color="LightBlue"):
> G4:=plot([[-1.75,-2],[1.75,-2]],[[0,-1.8],[0,-5.5]]],
  color=black,thickness=3):
> G4a:=plot([[-2,-2],[2,-2]],[[-2,-3],[2,-3]],[[-2,-4],[2,-4]],
  [[-2,-5],[2,-5]],[[-2,-2],[2,-5.4]],[[-1,-2],[2,-5.4]],
  [[1,-2],[2,-5.4]],[[2,-2],[2,-5.4]]],color=black):
> G5:=textplot([[-1.47,4.4,"T_1"],[1.48,3.5,"T_2"],
  [-0.5,3.5,"A_B"],[0.5,4,"1 Qm"],[0.5,2,"W_A"]],
  font=[HELVETICA,BOLD,12],color=black):
> display({G0,G1,G2,G3,G3a,G5},labels=["x [m]","y [m]"]);

```

Obr. 1 Stanovení plochy pracovního záběru



Plochu záběru W_A vypočteme jako rozdíl plochy vymezené osou x a červenou křivkou – dráha 1. prstu a plochy A_B vymezené osou x a modrou křivkou – dráha 4. prstu. Od výsledku je ještě nutné odečíst plochy lichoběžníků T_1 a T_2 , které jsou vymezeny osou x a počátkem, resp. koncem záběru. Pro zvýšení přehlednosti je přikreslena plocha 1 m^2 , která má tvar obdélníku, protože byla použita rozdílná měřítka pro osu x a y .

> **A_R:=Int(P1[2]*Diff(P1[1],t),t=t1..t1+tau);** *Plocha pod červenou křivkou*

$$A_R := \int_{-1.310559471}^{-1.310559471 + \tau} \left(-\frac{10t}{3} - \frac{3}{2} \cos(2\pi t) + \frac{3}{5} \sin(2\pi t) \right) \left(\frac{\partial}{\partial t} \left(\frac{3}{2} \sin(2\pi t) + \frac{3}{5} \cos(2\pi t) \right) \right) dt$$

> **A_R:=evalf(value(A_R));**

$$\begin{aligned} A_R := & 7.507578739 + 1.825344226 \cos(-8.234488014 + 6.283185308 \tau) \\ & + 6.871107239 \sin(-8.234488014 + 6.283185308 \tau) \\ & - 4.999999999 \sin(-8.234488014 + 6.283185308 \tau) \tau \\ & - 2.000000000 \cos(-8.234488014 + 6.283185308 \tau) \tau \\ & - 0.9449999999 \cos(-8.234488014 + 6.283185308 \tau) \sin(-8.234488014 + 6.283185308 \tau) \\ & - 8.199556825 \tau - 0.9000000000 \cos(-8.234488014 + 6.283185308 \tau)^2 \end{aligned}$$

> **A_B:=Int(P4[2]*Diff(P4[1],t),t=t1..t1+tau);** *Plocha pod modrou křivkou*

$$A_B := \int_{-1.310559471}^{-1.310559471 + \tau} \left(-\frac{10t}{3} - \frac{6}{5} \cos(2\pi t) + \frac{3}{5} \sin(2\pi t) \right) \left(\frac{\partial}{\partial t} \left(\frac{6}{5} \sin(2\pi t) + \frac{3}{5} \cos(2\pi t) \right) \right) dt$$

> **A_B:=evalf(value(A_B));**

$$\begin{aligned} A_B := & 6.185380725 + 1.984499169 \cos(-8.234488014 + 6.283185308 \tau) \\ & + 5.560547770 \sin(-8.234488014 + 6.283185308 \tau) \\ & - 4.000000001 \sin(-8.234488014 + 6.283185308 \tau) \tau \\ & - 2.000000000 \cos(-8.234488014 + 6.283185308 \tau) \tau \\ & - 0.5399999999 \cos(-8.234488014 + 6.283185308 \tau) \sin(-8.234488014 + 6.283185308 \tau) \\ & - 5.654866778 \tau - 0.7199999999 \cos(-8.234488014 + 6.283185308 \tau)^2 \end{aligned}$$

> **T_1:=evalf(subs(t=t1,(P1[2]+P4[2])*(P4[1]-P1[1])/2));**

Plocha levého lichoběžníku je konstantní

$$T_1 := 1.201306679$$

> **T_2:=evalf(subs(t=t1+tau,(P1[2]+P4[2])*(P1[1]-P4[1])/2));**

Plocha pravého lichoběžníku – funkcí času

$$\begin{aligned} T_2 := & 0.1500000000 (8.737063140 - 6.666666667 \tau - 2.700000000 \cos(-8.234488014 + 6.283185308 \tau) \\ & + 1.200000000 \sin(-8.234488014 + 6.283185308 \tau)) \sin(-8.234488014 + 6.283185308 \tau) \end{aligned}$$

> **W_A:=A_B+T_1+T_2-A_R;**

> **W_AF:=evalf(subs(tau=d_t,W_A));** *celková plocha jednoho záběru*

$$W_{AF} := 1.390562352$$

Pro ilustraci je možné vykreslit závislosti jednotlivých dílčích ploch na čase, viz Obr. 2.


```
> display({plot([A_R/5,A_B/5,T_1/5,T_2/5,W_A],tau=0..d_t,
color=[red,blue,black,black,"DarkGreen"],thickness=[2,2,2,2,3]),
textplot(subs(tau=d_t*0.95,[tau,A_R/5,"A_R/5"]),color=red,
align=LEFT,font=[HELVETICA,12]),
textplot(subs(tau=(t2-t1)*0.9,[tau,A_B/5,"A_B/5"]),color=blue,
align=LEFT,font=[HELVETICA,12]),textplot(subs(tau=(t2-t1)*0.75,
```

```

[tau,W_A,"W_A"],color="DarkGreen",align=RIGHT,
font=[HELVETICA,BOLD,12]),
textplot(subs(tau=(t2-t1)*0.75,[tau,T_2/5,"T_2"]),color=black,
align={LEFT,ABOVE},font=[HELVETICA,12]),
textplot(subs(tau=(t2-t1)*0.75,[tau,1.1*T_1/5,"T_1"]),
color=black,align={LEFT,ABOVE},font=[HELVETICA,12]),
labels=["t [s]","A [Qm]"]);

```

Obr. 2 Závislosti jednotlivých ploch na čase




Dále je možné vykreslit graf, zobrazující velikost plochy záběru a její derivaci podle času, viz Obr. 3.

```

> dW_A:=diff(W_A,tau):
> display({plot([W_A,dW_A],tau=0..d_t,labels=["t [s]",""],
color=["DarkGreen",red],thickness=3),
textplot([0.02,0.5,"W_A [Qm]"],color="DarkGreen",
font=[HELVETICA,BOLD,12],align=RIGHT),
textplot([0.02,3,"dW_A/dt [Qm/s]"],color=red,
font=[HELVETICA,BOLD,12],align=RIGHT)});

```


Obr. 3 Časová závislost plochy záběru a její derivace podle času



Kinematika středového bodu pracovního záběru

Za středový bod pracovního záběru zvolíme střed úsečky tvořené prvním a čtvrtým prstem. Pro dynamické výpočty je nutné nejprve znát kinematické charakteristiky středového bodu. Těmito charakteristikami jsou:

Název veličiny	Označení v Maple
Polohový vektor středového bodu, dále <i>SB</i>	T_c
Vektor rychlosti <i>SB</i>	V_c
Absolutní velikost vektoru rychlosti <i>SB</i>	VA
Vektor zrychlení <i>SB</i>	A
Absolutní velikost vektoru zrychlení <i>SB</i>	AA
Tečné zrychlení <i>SB</i>	At
Normálové zrychlení <i>SB</i>	An
Složka <i>x</i> vektoru rychlosti <i>SB</i>	V_x
Složka <i>y</i> vektoru rychlosti <i>SB</i>	V_y
Složka <i>x</i> vektoru zrychlení <i>SB</i>	A_x
Složka <i>y</i> vektoru zrychlení <i>SB</i>	A_y

```
> T_c:=evalf(expand(subs(t=t1+tau,(P1+P4)/2))):
> V_c:=collect(diff(T_c,tau),[sin,cos]):
> VA:=combine(expand(sqrt(V_c[1]^2+V_c[2]^2))):
> A_c:=collect(simplify(diff(T_c,tau,tau)),[sin,cos]):
> AA:=combine(expand(sqrt(A_c[1]^2+A_c[2]^2))):
```

```

> At:=evalf(diff(VA,tau)):
> An:=evalf(sqrt(AA^2-At^2)):
> V_x:=V_c[1]; V_y:=V_c[2]:
> A_x:=A_c[1]; A_y:=A_c[2]:

```


Nyní je možné vykreslit graf znázorňující průběh vektoru rychlosti středového bodu, Viz Obr. 4.

```

> display({plot([V_x,V_y,VA],tau=0..d_t,color=[red,blue,black],
  labels=["t [s]","V [m/s]"],thickness=[2,2,3]),
  textplot(subs(tau=0.8*d_t,[tau,VA,"|V|"]),
  font=[HELVETICA,BOLD,12],color=black,align={ABOVE,RIGHT}),
  textplot(subs(tau=0.9*d_t,[tau,V_x,"Vx"]),
  font=[HELVETICA,BOLD,12],color=red,align=RIGHT),
  textplot(subs(tau=0.7*d_t,[tau,V_y,"Vy"]),
  font=[HELVETICA,BOLD,12],color=blue,align=RIGHT)});

```

Obr. 4 Rychlostní charakteristiky středového bodu




Délku dráhy pracovního bodu v závislosti na čase je možné vykreslit jako integrál absolutní velikosti rychlosti podle času, viz Obr. 5.

```

> plot(evalf(Int(VA,tau=0..tau)),tau=0..d_t,thickness=3,
  labels=["t [s]","d [m]"]);

```


Obr. 5 Délka trajektorie pracovního bodu v závislosti na čase



Na závěr této části je možné vykreslit časový průběh vektoru zrychlení pracovního bodu, viz. Obr.6.

```
> display({plot([AA,At,An,A_x,A_y],tau=0..d_t,
  color=[black,red,blue,green,brown],
  thickness=3,labels=["t [s]","A [m/s2]"]),
  textplot(subs(tau=d_t/3,[tau,AA*1.02,"|A|"]),align=ABOVE,
  font=[HELVETICA,BOLD,12],color=black),
  textplot(subs(tau=d_t/2,[tau,At*0.95,"At"]),align=ABOVE,
  font=[HELVETICA,BOLD,12],color=red),
  textplot(subs(tau=d_t/2,[tau,An*0.9,"An"]),align=ABOVE,
  font=[HELVETICA,BOLD,12],color=blue),
  textplot(subs(tau=d_t/3,[tau,A_x,"Ax"]),align=RIGHT,
  font=[HELVETICA,BOLD,12],color=green),
  textplot(subs(tau=d_t/8,[tau,A_y,"Ay"]),align=RIGHT,
  font=[HELVETICA,BOLD,12],color=brown)});
```

Obr. 6 Časový průběh vektoru zrychlení pracovního bodu



Výpočet výkonu potřebného v průběhu pracovního záběru

Celková spotřeba energie pro jeden pracovní záběr je součtem předané kinetické energie a práce třecí síly. Je důležité si uvědomit, že jde o pohyb tělesa s proměnnou hmotností. Na začátku záběru je množství píce nesené prsty nulové a s přírůstkem plochy záběru se zvyšuje. Energetickou bilanci záběru udává rovnice eq1.

```
> eq1 := E(t) = m(t) * (Vx(t)^2 + Vy(t)^2) / 2 +
      Int(m(t) * g * mu * sqrt(Vx(t)^2 + Vy(t)^2), t);
```

$$eq1 := E(t) = \frac{1}{2} m(t) (Vx(t)^2 + Vy(t)^2) + \int m(t) g \mu \sqrt{Vx(t)^2 + Vy(t)^2} dt$$

Derivací této rovnice a využitím zjednodušujících substitucí je možné vypočíst potřebný výkon.

```
> eq2 := map(u->diff(u,t), eq1);
```

$$eq2 := \frac{d}{dt} E(t) = \frac{1}{2} \left(\frac{d}{dt} m(t) \right) (Vx(t)^2 + Vy(t)^2) + \frac{1}{2} m(t) \left(2 Vx(t) \left(\frac{d}{dt} Vx(t) \right) + 2 Vy(t) \left(\frac{d}{dt} Vy(t) \right) \right) + m(t) g \mu \sqrt{Vx(t)^2 + Vy(t)^2}$$

```
> DSu:=[diff(E(t),t)=P(t),diff(m(t),t)=sigma*dS(t),
  m(t)=sigma*S(t),diff(Vx(t),t)=Ax(t),
  diff(Vy(t),t)=Ay(t),Vx(t)^2+Vy(t)^2=V(t)^2];
DSu := [ d/dt E(t) = P(t), d/dt m(t) = sigma dS(t), m(t) = sigma S(t), d/dt Vx(t) = Ax(t), d/dt Vy(t) = Ay(t),
  Vx(t)^2 + Vy(t)^2 = V(t)^2 ]
```

```
> eq3:=expand(simplify(subs(DSu,eq2),symbolic));
eq3 := P(t) = 1/2 sigma dS(t) V(t)^2 + sigma S(t) Vx(t) Ax(t) + sigma S(t) Vy(t) Ay(t) + sigma S(t) g mu V(t)
```

Výsledný výkon je možné rozdělit na tři rozdílné členy.

> **P_1=select(has,rhs(eq3),dS);** Výkon udělující rychlost V přibírané hmotě

$$P_1 = \frac{1}{2} \sigma dS(t) V(t)^2$$

> **P_2=select(has,rhs(eq3),mu);** Výkon třecí síly, který závisí na koeficientu tření $-\mu$. Tento výkon udržuje nabranou hmotu v pohybu.

$$P_2 = \sigma S(t) g \mu V(t)$$

> **P_3=factor(remove(has,rhs(eq3),[dS,mu]));** Výkon měnící vektor rychlosti již nabrané pohybující se hmoty

$$P_3 = \sigma S(t) (Vx(t) Ax(t) + Vy(t) Ay(t))$$

```
> P_1:=subs(KSu,sigma)*dW_A*VA^2:
```


```
> P_2:=subs(KSu,sigma*g)*W_A*mu*VA:
```

```
> P_3:=subs(KSu,sigma*g)*W_A*(V_x*A_x+V_y*A_y):
```

Tyto tři dílčí výkony i jejich výsledný součet je možné znázornit graficky, viz Obr. 7.

```
> display({plot(P_1,tau=0..d_t,color=red,thickness=3),
  plot(P_3,tau=0..d_t,thickness=2,color=blue),
  plot([seq(P_2,mu=[0.5,1])],tau=0..d_t,color=brown,thickness=2),
  plot([seq(P_1+P_2+P_3,mu=[0.5,1])],tau=0..d_t,
  color=black,thickness=3),
  textplot(subs(tau=0.15,[tau*1.1,P_1,"P_1"]),
  font=[HELVETICA,BOLD,12],color=red,align={ABOVE,RIGHT}),
  textplot(subs(tau=0.2,mu=1,[tau,P_2*1.5,"P_2,mu=[0.5,1.0]"]),
  font=[HELVETICA,BOLD,12],color=brown,align=ABOVE),
  textplot(subs(tau=0.3,[tau,P_3*0.98,"P_3"]),
  font=[HELVETICA,BOLD,12],color=blue,align=ABOVE)},
  textplot(subs(tau=0.2,mu=1,[tau*1.05,P_1+P_2+P_3,"P(t),
  mu=[0.5,1.0]"]),font=[HELVETICA,BOLD,12],
  color=black,align=RIGHT),labels=["t [s]", "P [W]"]);
```

Obr. 7 Celkový potřebný výkon a jeho skladba



Ze záporných hodnot celkového výkonu označeného černou křivkou a výkonu P_3 označeného modrou křivkou je možné konstatovat:

- 1) Píce je urychlena z klidu na rychlost posunu, tomu odpovídá kladná část černé křivky.
- 2) Rychlost posunu klesá, tomu odpovídá i záporné tečné zrychlení z Obr. 6. Protože rychlost již nabrané píce je vyšší, takže prst ji netlačí protože letí vzduchem před ním. Brzdí ji jen odpor vzduchu, který lze zanedbat.
- 3) Výkon třecí síly lze zanedbat oproti ostatním i pro velké koeficienty μ . Na Obr. 7. jsou vykresleny křivky pro $\mu=0,5$ a $\mu=1,0$, což je již skutečně vysoká hodnota koeficientu tření. Při této velikosti μ je již na vlečení tělesa nutná síla rovná tíze tělesa!
- 4) Jediný nezanedbatelný výkon konající práci je P_1 označený červenou křivkou. Tento výkon urychluje ležící píci z klidu na rychlost posunu prstu.
- 5) Proto se dále bude uvažovat pouze výkon P_1 . Ostatní výkony buď nemají význam, protože nekonají práci, nebo je lze zanedbat

V dalším kroku vypočteme střední hodnotu výkonu P_1 v průběhu jednoho pracovního záběru. Dále vykreslíme graf, ve kterém bude naznačen průběh tohoto výkonu a jeho střední hodnota. Je nutné si uvědomit, že v průběhu záběru prvního ramene předcházející ramena v záběru již jsou a další ramena do záběru nastupují, ještě než první rameno svůj záběr dokončí. Tyto skutečnosti jsou dobře patny z grafu na Obr. 8.

> $P_S := 1 / 'd_t' * \text{Int}('P_1', \text{tau} = 0 .. 'd_t')$


$$P_S := \frac{1}{d_t} \int_0^{d_t} P_1 dt$$

```

> P_S:=evalf(P_S);
                                P_S := 4425.325592
> display([seq(plot(subs(tau=tau-j*d_t/6,P_1),
tau=j*d_t/6..d_t),j=1..5),seq(plot(subs(tau=tau+j*d_t/6,P_1),
tau=0..d_t-j*d_t/6),j=1..5),
plot(P_1,tau=0..d_t,thickness=3),
plot(P_S,tau=0..d_t,color=black,thickness=5),
textplot([seq([0.05+j*d_t/6,8500,cat(convert(j+1,string),
". shoulder")],j=0..5)],font=[HELVETICA,BOLD,12],color=red),
textplot([seq([0.095+j*d_t/6,700,cat(convert(8+j,string),
". shoulder")],j=0..4)],font=[HELVETICA,BOLD,12],color=red),
textplot([0.25,4700,"Mean Power of one shoulder = 4.5 kW -> Total
Power = 27 kW"],
font=[HELVETICA,BOLD,15],color=black)],labels=["t [s]","P [W]"]);

```

Obr. 8 Výkon prvního ramene a současné výkony dalších ramen



Z Grafu na Obr. 8. je zcela jasně patrné, že celkový činný výkon stroje je 6x střední výkon prvního ramene, tedy cca **27 kW!**

Dynamika záběru


Jsou li výkony P_2 a P_3 zanedbatelné, musí být zanedbatelné i odpovídající síly. Sílu F_I odpovídající P_I je možné stanovit ze vztahu $P_I = |F_I(t)| |V(t)| \cos(\alpha)$. Protože F_I a V jsou rovnoběžné, síla F_I působí přesně ve směru pohybu, tedy vektoru V , pak $\alpha = 0$ a $\cos(\alpha) = 1$. Výpočet absolutní velikosti F_I je pak jednoduchý a směr vektoru síly F_I je rovnoběžný s vektorem rychlosti V . Dále vypočteme maximální silové namáhání a stanovíme okamžik, kdy nastává. Na závěr vypočteme moment síly, která ohybá prst a celkový moment síly působící na všechny čtyři prsty.

```

> F_1:=subs(KSu,sigma)*dW_A*VA: F_1x:=subs(KSu,sigma)*dW_A*V_x:
> F_1y:=subs(KSu,sigma)*dW_A*V_y:
> t_F_1max:=fsolve(diff(F_1,tau),tau=0.02);
> F_1max:=evalf(subs(tau=t_F_1max,F_1));
      t_F_1max := 0.03336439077   F_1max := 664.0589152
> MoF_1:=subs(KSu,1/n)*F_1max;
      MoF_1 := 66.40589152
> display({plot([F_1,F_1x,F_1y],tau=0..d_t,thickness=3,
  labels=["t [s]","F [N]"],color=[black,red,blue]),
  textplot([0.09,200,"F_x"],font=[HELVETICA,BOLD,12],color=red),
  textplot([0.09,-500,"F_y"],font=[HELVETICA,BOLD,12],color=blue),
  textplot([0.09,580,"|F|"],font=[HELVETICA,BOLD,12],color=black),
  textplot([t_F_1max,680,"Fmax=670N"],font=[HELVETICA,BOLD,12],
  color=black,align=ABOVE),
  textplot([t_F_1max,-20,"t=33ms"],font=[HELVETICA,BOLD,12],
  color=black,align=BELOW),
  plot([t_F_1max,0],[t_F_1max,F_1max],color=black,thickness=1)});

```

Obr. 9 Časový průběh vektoru síly a maximum velikosti síly



VÝSLEDKY A DISKUZE

Popisovaný výpočet dynamiky shrnovače píce Fella TS 1602 Hydro navazuje na výpočty, které provedli Šmíd a Bartoň (2008) a tyto výpočty dále rozšiřuje.

Umožňuje stanovit základní dynamické parametry, ale také potřebné výkony nejen pro pohyb samotného stroje, ale i pro jeho vlastní práci, tj. hnutí posečené píce po strništi. Při vnikání pracovních prstů do shrnovaného řádku dochází k narůstání odebraného výkonu vlivem přibývajícího objemu shrnované píce, ovšem v tu samou chvíli již jiné pracovní prsty materiál hromadí a další z nich daný řádek opouští, a tím se jejich energetický požadavek snižuje. Tudíž se celkový odebraný výkon drží na zhruba stálé hladině.

Výkonnostní požadavky plně korespondují s potřebným výkonem tažného prostředku, který udává výrobce.

Z výpočtů je možné zjistit i namáhání konců pracovních prstů které v okamžiku vstupu do posečeného materiálu je mžikově dosti extrémní. Tyto prudké skokové změny v namáhání kladou velmi vysoké požadavky na jakost materiálu i na jejich upevnění ke shrnovacímu ramenu.

ZÁVĚR

Výše uvedený postup umožňuje provést všechny výpočty potřebné ke stanovení dynamických charakteristik základních pracovních částí stroje, jejich energetických požadavků, i vybraných sil, které na ně působí. Z důvodů úspory místa a zvýšení přehlednosti nebyly uvedeny všechny výstupy programu. Nicméně je zcela zřejmé, že při použití výše uvedených postupů je možné výrazně zefektivnit vývojové a konstrukční práce spojené s projektováním nových typů zemědělských mechanismů.

LITERATURA

ŠMÍD, V. *Žací stroje s úpravou pokosu*. Brno : MZLU Brno, 2007. Vedoucí diplomové práce doc. Ing. Jan Červinka, CSc.

ŠMÍD, V., BARTOŇ, S. *Matematický model shrnovače píce FELLA TS 1602*, In *MendelNet'08 Agro*, listopad 2008, MZLU Brno, s. 82, ISBN 987-80-7375-239-2

HAKL, Z. *Matematické modelování biologických a technologických procesů v zemědělství*. Disertační práce. MZLU: MZLU Brno, 2005. 95 s.

BARTOŇ, S. *The Kinematics of Agricultural Machines*.<http://www.mapleapps.com/categories/engineering/mechanical/html/agriculture.html>

JAVOREK, F. *Technika pro senážování*, In. *Zemědělec*, březen 2008, č. 11, s. 24 – 30

ČERVINKA, J. *Stroje pro sklizeň píce na seno*. 2. upravené vyd. Praha : ÚZPI, 2002. ISBN 80-7105-054-7

MAPLESOFT. *Maple10 – User Manual*. Waterloo Maple, 2005, ISBN 1-894511-75-1

Propagační materiály firmy FELLA